```
// ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
funcprot(0);
// ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

gamma=6.668E-11; // Gravitationskonstante

Ms=1.99E+30; // Masse der Sonne
Me=5.98E+24; // Masse der Erde
Mm=7.35E+22; // Masses des Mondes

Re=6.37E+6; // Radius der Erde

// ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

function longh=flongh(longf)
    if longf>%pi then
        longh=longf-2*%pi;
    else
        longh=longf;
    end
endfunction

function lath=flath(latf)
    if latf>%pi/2 then
        lath=latf-%pi;
    else
        lath=latf;
    end
endfunction

// ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

function String(x, y, text, col, sz, style)
  xstring(x,y,text);
  e=gce();
  e.font_foreground = col;
  e.font_size = sz;
  e.font_style = style;
endfunction

function plot_mark(sph, col, fg, sz, name)
  plot(sph(1)*%CL_rad2deg, sph(2)*%CL_rad2deg, "o");

  e=gce();
  e.children.mark_background=col;
  e.children.mark_foreground=fg;
  e.children.mark_size=sz;

  String(sph(1)*%CL_rad2deg, sph(2)*%CL_rad2deg, " "+name, col,3,5);
endfunction

// ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
// main
// ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
```

```
// longitude/latitude of location
lon=11.42*%CL_deg2rad;
lat=48.15*%CL_deg2rad;
alt=531;

// scal = reference date (string, calendar)
scal = [];

desc_param = list(..
   CL_defParam("Longitude of location", lon, units=["rad", "deg"],..
               valid='$x>=0 & $x <=360'), ..
   CL_defParam("Latitude of location", lat, units=["rad", "deg"],..
               valid='$x>=-90 & $x <=90'), ..
   CL_defParam("Reference time (calendar format, TREF)", scal, typ="cal") ..
);

[lon,lat,scal] = CL_inputParam(desc_param);

loc_ell = [lon;lat;0];

t0 = CL_dat_cal2cjd(CL_dat_str2cal(scal));

cal = CL_dat_cjd2cal(t0);

months=["Jan","Feb","Mär","Apr","Mai","Jun","Jul","Aug","Sep",..
        "Okt","Nov","Dez"];
str_day=sprintf("%d %s %d", cal(3), months(cal(2)), cal(1));
str_hour=sprintf("%02d:%02d", cal(4), cal(5));
dateText=str_day+" "+str_hour+" (TREF)";

// ++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

loc_car=CL_co_ell2car(loc_ell);

pos_sun_ECI = CL_eph_sun(t0)
pos_sun_ECI_sph=CL_co_car2sph(pos_sun_ECI);

pos_sun_ECF=CL_fr_convert("ECI","ECF",t0,pos_sun_ECI);
pos_sun_ECF_sph=CL_co_car2sph(pos_sun_ECF);
pos_sun_ECF_sph(1)=flongh(pos_sun_ECF_sph(1));
pos_sun_ECF_sph(2)=flath(pos_sun_ECF_sph(2));

pos_moon_ECI = CL_eph_moon(t0)
pos_moon_ECI_sph=CL_co_car2sph(pos_moon_ECI);

pos_moon_ECF=CL_fr_convert("ECI","ECF",t0,pos_moon_ECI);
pos_moon_ECF_sph=CL_co_car2sph(pos_moon_ECF);
pos_moon_ECF_sph(1)=flongh(pos_moon_ECF_sph(1));
pos_moon_ECF_sph(2)=flath(pos_moon_ECF_sph(2));

pos_moonBK_ECF_sph=zeros(pos_moon_ECF_sph);
pos_moonBK_ECF_sph(1)=flongh(%pi + pos_moon_ECF_sph(1));
pos_moonBK_ECF_sph(2)=flath(%pi/2 + pos_moon_ECF_sph(2));

disp(pos_sun_ECF_sph(1)*%CL_rad2deg);
```

```
disp(pos_sun_ECF_sph(2)*%CL_rad2deg);

disp(pos_moon_ECF_sph(1)*%CL_rad2deg);
disp(pos_moon_ECF_sph(2)*%CL_rad2deg);

disp(pos_moonBK_ECF_sph(1)*%CL_rad2deg);
disp(pos_moonBK_ECF_sph(2)*%CL_rad2deg);

col1 = color("darkorange"); // Sonne
col2 = color("maroon"); // Mond
col4 = color("violet"); // Mond Gegenpunkt
col3 = color("scilabgreen3"); // Ort

f=scf();
f.event_handler_enable="off";
Col = addcolor([0.6,0.7,0.8]);

f.pixmap= "on";
f.immediate_drawing="off";
CL_plot_earthMap(color_id=Col);

plot_mark(pos_sun_ECF_sph,col1,1,10,"-S");
plot_mark(pos_moon_ECF_sph,col2,1,10,"-M");
plot_mark(pos_moonBK_ECF_sph,col4,1,10,"-M");
plot_mark(loc_ell,col3,1,10,"-O");

a=gca();
a.x_label.text = 'Länge [°]';
a.y_label.text = 'Breite [°]';
a.title.text = 'Sonnen- und Mondrichtung  -  ' + dateText;

CL_g_stdaxes(a);

f.immediate_drawing="on";
f.pixmap= "off";
```